
	Reason for change:
	1) During TCP connection establishment by SS,  if the UE has any ongoing the TCP connection close , ongoing TCP conneciton will be close can occur before new TCP connection establishement occurs. To tackle this issue, an boolean flag tcpClientConnectInd  is passed  to f_TcpClient_Connect so that connection close can be determined.
2)  As described in the change 1, TCP connection termination needs to be handling during the TCP connection establishment.
3) As defined in change 1, flag , tcpClientConnectInd needs to be passed. It is passed as ‘inout’ parameter as this value needs to be retained by calling function

	
	

	Summary of change:
	1) Defined flag tcpClientConnectInd and passed onto f_TcpClient_Connect. 
2) Defined alt statement to handle TCP conneciton termination

3) Passed the flag tcpClientConnectInd to called function fl_TCP_Connect

	
	

	Consequences if not approved:
	A comformant UE may fail the test.

Change 1
	Function name
	fl_IP_IMS_DL_Send

	Reason for change
	During TCP connection establishment by SS,  if the UE has any ongoing the TCP connection close , ongoing TCP conneciton will be close can occur before new TCP connection establishement occurs. To tackle this issue, an boolean flag tcpClientConnectInd  is passed  to f_TcpClient_Connect so that connection close can be determined.

	Summary of change
	Defined flag tcpClientConnectInd and passed onto f_TcpClient_Connect.

	Source of change
	IP_PTC\IP_PTC_IMS_Handler.ttcn


Before change:

	       function fl_IP_IMS_DL_Send(inout IMS_Server_Type p_ImsServer,

                             IMS_RoutingInfo_Type p_RoutingInfo,

                             IMS_RequestResponse_Type p_RequestResponse,

                             bitstring p_Data) runs on IP_PTC

  {

    var InternetProtocol_Type v_Protocol := p_RoutingInfo.Protocol;

    var IMS_ProtectedUnprotected_Type v_ProtectedUnprotected;

    var IMS_LocalRemotePort_Type v_LocalRemotePort;

    var template (omit) PortNumber_Type v_LocalPort;

    var PortNumber_Type v_RemotePort;

    var IP_AddrInfo_Type v_UE_Address;

    var IP_AddrInfo_Type v_NW_Address;

    var template (value) IP_Socket_Type v_LocalSocket;

    var template (value) IP_Socket_Type v_RemoteSocket;

    var template (omit) IP_Connection_Type v_TCP_Connection := omit;

    var Datagram_Content_Type v_Data := bit2oct(p_Data);

    var boolean v_SecurityIsEstablished := fl_ImsServer_SecurityIsEstablished(p_ImsServer);

    if (isvalue(p_RoutingInfo.Security)) {

      /* Security may be omitted by the IMS PTC when the IMS message shall be sent protected or unprotected based on whether security is configured or not;

         in case of the initial REGISTER response security is already configured but the reponse needs to be sent unprotected (as the UE has not fully configured security yet)

         => at least in this case the IMS PTC shall set the "Security" field of the RoutingInfo */

      v_ProtectedUnprotected := p_RoutingInfo.Security;

      if ((v_ProtectedUnprotected == protected) and not v_SecurityIsEstablished) {

        FatalError(__FILE__, __LINE__, "IMS security not established");

      }

    } else if (v_SecurityIsEstablished) {

      v_ProtectedUnprotected := protected;

    } else {

      v_ProtectedUnprotected := unprotected;

    }

    v_LocalRemotePort := fl_IP_IMS_GetPortTX(p_ImsServer, v_ProtectedUnprotected, v_Protocol, p_RequestResponse);

    v_LocalPort  := v_LocalRemotePort.Local;

    v_RemotePort := v_LocalRemotePort.Remote;

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {

      v_NW_Address := p_RoutingInfo.NW_Address;

      v_UE_Address := p_RoutingInfo.UE_Address;

    } else if (fl_ImsServer_UeIsRegistered(p_ImsServer)) {

      v_NW_Address := p_ImsServer.RegistrationInfo.NW_Address;

      v_UE_Address := p_ImsServer.RegistrationInfo.UE_Address;

    } else {

      FatalError(__FILE__, __LINE__, "missing IP addresses");

    }

    select (v_Protocol) {

      case (udp) {

        v_LocalSocket  := cs_IP_Socket(v_NW_Address, v_LocalPort);

        v_RemoteSocket := cs_IP_Socket(v_UE_Address, v_RemotePort);

        IP_SOCK.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_Data));

      }

      case (tcp) {

        select (v_ProtectedUnprotected) {

          case (unprotected) {

            select (p_RequestResponse) {

              case (response) {

                v_TCP_Connection := p_ImsServer.TcpConnections.Unprotected_uc_ps;

              }

              case (request) {

                v_TCP_Connection := p_ImsServer.TcpConnections.Unprotected_pc_us;

                if (not isvalue(v_TCP_Connection)) {

                  v_LocalPort := omit;     // in case of an unprotected TCP connection to UE the is no need to specify any specific local port but we can use an ephemeral port instead

                  v_TCP_Connection := f_TcpClient_Connect(cs_IP_Socket(v_NW_Address, v_LocalPort), cs_IP_Socket(v_UE_Address, v_RemotePort), p_ImsServer.DrbInfo, ims);
                  p_ImsServer.TcpConnections.Unprotected_pc_us := valueof(v_TCP_Connection);

                }

              }

            }

          }

          case (protected) {

            select (p_RequestResponse) {

              case (response) {

                v_TCP_Connection := p_ImsServer.TcpConnections.Protected_uc_ps;

              }

              case (request) {

                v_TCP_Connection := p_ImsServer.TcpConnections.Protected_pc_us;

                if (not isvalue(v_TCP_Connection)) {

                  v_TCP_Connection := f_TcpClient_Connect(cs_IP_Socket(v_NW_Address, v_LocalPort), cs_IP_Socket(v_UE_Address, v_RemotePort), p_ImsServer.DrbInfo, ims);
                  p_ImsServer.TcpConnections.Protected_pc_us := valueof(v_TCP_Connection);

                }

              }

            }

          }

        }

        if (not isvalue(v_TCP_Connection)) {

          FatalError(__FILE__, __LINE__, "no valid TCP connection");

        }

        IP_SOCK.send(cs_TCP_DATA_REQ(valueof(v_TCP_Connection), v_Data));

      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid protocol");

        return;

      }

    }

  }



After change :

	function fl_IP_IMS_DL_Send(inout IMS_Server_Type p_ImsServer,

                             IMS_RoutingInfo_Type p_RoutingInfo,

                             IMS_RequestResponse_Type p_RequestResponse,

                             bitstring p_Data) runs on IP_PTC

  {

    var InternetProtocol_Type v_Protocol := p_RoutingInfo.Protocol;

    var IMS_ProtectedUnprotected_Type v_ProtectedUnprotected;

    var IMS_LocalRemotePort_Type v_LocalRemotePort;

    var template (omit) PortNumber_Type v_LocalPort;

    var PortNumber_Type v_RemotePort;

    var IP_AddrInfo_Type v_UE_Address;

    var IP_AddrInfo_Type v_NW_Address;

    var template (value) IP_Socket_Type v_LocalSocket;

    var template (value) IP_Socket_Type v_RemoteSocket;

    var template (omit) IP_Connection_Type v_TCP_Connection := omit;

    var Datagram_Content_Type v_Data := bit2oct(p_Data);

    var boolean v_SecurityIsEstablished := fl_ImsServer_SecurityIsEstablished(p_ImsServer);

    var boolean tcpClientConnectInd := false;
    if (isvalue(p_RoutingInfo.Security)) {

      /* Security may be omitted by the IMS PTC when the IMS message shall be sent protected or unprotected based on whether security is configured or not;

         in case of the initial REGISTER response security is already configured but the reponse needs to be sent unprotected (as the UE has not fully configured security yet)

         => at least in this case the IMS PTC shall set the "Security" field of the RoutingInfo */

      v_ProtectedUnprotected := p_RoutingInfo.Security;

      if ((v_ProtectedUnprotected == protected) and not v_SecurityIsEstablished) {

        FatalError(__FILE__, __LINE__, "IMS security not established");

      }

    } else if (v_SecurityIsEstablished) {

      v_ProtectedUnprotected := protected;

    } else {

      v_ProtectedUnprotected := unprotected;

    }

    v_LocalRemotePort := fl_IP_IMS_GetPortTX(p_ImsServer, v_ProtectedUnprotected, v_Protocol, p_RequestResponse);

    v_LocalPort  := v_LocalRemotePort.Local;

    v_RemotePort := v_LocalRemotePort.Remote;

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {

      v_NW_Address := p_RoutingInfo.NW_Address;

      v_UE_Address := p_RoutingInfo.UE_Address;

    } else if (fl_ImsServer_UeIsRegistered(p_ImsServer)) {

      v_NW_Address := p_ImsServer.RegistrationInfo.NW_Address;

      v_UE_Address := p_ImsServer.RegistrationInfo.UE_Address;

    } else {

      FatalError(__FILE__, __LINE__, "missing IP addresses");

    }

    select (v_Protocol) {

      case (udp) {

        v_LocalSocket  := cs_IP_Socket(v_NW_Address, v_LocalPort);

        v_RemoteSocket := cs_IP_Socket(v_UE_Address, v_RemotePort);

        IP_SOCK.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_Data));

      }

      case (tcp) {

        select (v_ProtectedUnprotected) {

          case (unprotected) {

            select (p_RequestResponse) {

              case (response) {

                v_TCP_Connection := p_ImsServer.TcpConnections.Unprotected_uc_ps;

              }

              case (request) {

                v_TCP_Connection := p_ImsServer.TcpConnections.Unprotected_pc_us;

                if (not isvalue(v_TCP_Connection)) {

                  v_LocalPort := omit;     // in case of an unprotected TCP connection to UE the is no need to specify any specific local port but we can use an ephemeral port instead

                  v_TCP_Connection := f_TcpClient_Connect(cs_IP_Socket(v_NW_Address, v_LocalPort), cs_IP_Socket(v_UE_Address, v_RemotePort), p_ImsServer.DrbInfo, ims, tcpClientConnectInd);
                  p_ImsServer.TcpConnections.Unprotected_pc_us := valueof(v_TCP_Connection);

                  if(tcpClientConnectInd) {

                  p_ImsServer.TcpConnections.Unprotected_uc_ps := omit;
                  }

                }

              }

            }

          }

          case (protected) {

            select (p_RequestResponse) {

              case (response) {

                v_TCP_Connection := p_ImsServer.TcpConnections.Protected_uc_ps;

              }

              case (request) {

                v_TCP_Connection := p_ImsServer.TcpConnections.Protected_pc_us;

                if (not isvalue(v_TCP_Connection)) {

                  v_TCP_Connection := f_TcpClient_Connect(cs_IP_Socket(v_NW_Address, v_LocalPort), cs_IP_Socket(v_UE_Address, v_RemotePort), p_ImsServer.DrbInfo, ims, tcpClientConnectInd);
                  p_ImsServer.TcpConnections.Protected_pc_us := valueof(v_TCP_Connection);

                  if(tcpClientConnectInd) {



   p_ImsServer.TcpConnections.Protected_uc_ps := omit;
                  }

                }

              }

            }

          }

        }

        if (not isvalue(v_TCP_Connection)) {

          FatalError(__FILE__, __LINE__, "no valid TCP connection");

        }

        IP_SOCK.send(cs_TCP_DATA_REQ(valueof(v_TCP_Connection), v_Data));

      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid protocol");

        return;

      }

    }

  }



Change 2

	Function name
	fl_TCP_Connect

	Reason for change
	As described in the change 1, TCP connection termination needs to be handling during the TCP connection establishment.

	Summary of change
	Defined alt statement to handle TCP conneciton termination

	Source of change
	IP_PTC\TCP_Functions.ttcn


Before change:

	  function fl_TCP_Connect(template (value) IP_Socket_Type p_LocalSocket,

                          template (value) IP_Socket_Type p_RemoteSocket,

                          InternetApplication_Type p_Application) runs on IP_PTC return IP_Socket_Type

  {

    var template (present) IP_Socket_Type v_LocalSocket := p_LocalSocket;

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    if (not isvalue(p_LocalSocket.Port)) {

      v_LocalSocket.Port := ?;

    }

    IP_SOCK.send(cs_TCP_CONNECT_REQ(p_LocalSocket, p_RemoteSocket, p_Application));

    IP_SOCK.receive(cr_TCP_CONNECT_CNF(v_LocalSocket, p_RemoteSocket)) -> value v_IP_SOCKET_IND;
    return v_IP_SOCKET_IND.CTRL.ConnectionId.Local;

  }


After change :

	function fl_TCP_Connect(template (value) IP_Socket_Type p_LocalSocket,

                          template (value) IP_Socket_Type p_RemoteSocket,

                          InternetApplication_Type p_Application,

                          inout boolean tcpClientConnectInd := false) runs on IP_PTC return IP_Socket_Type

  {

    var template (present) IP_Socket_Type v_LocalSocket := p_LocalSocket;

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    if (not isvalue(p_LocalSocket.Port)) {

      v_LocalSocket.Port := ?;

    }

    IP_SOCK.send(cs_TCP_CONNECT_REQ(p_LocalSocket, p_RemoteSocket, p_Application));

    //IP_SOCK.receive(cr_TCP_CONNECT_CNF(v_LocalSocket, p_RemoteSocket)) -> value v_IP_SOCKET_IND;
    alt

    {

       [] IP_SOCK.receive(cr_TCP_CONNECT_CNF(v_LocalSocket, p_RemoteSocket)) -> value v_IP_SOCKET_IND {

          }

       [] IP_SOCK.receive(cr_TCP_CLOSE_IND(?)) {

           tcpClientConnectInd := true;   

           repeat;

          }

    }
    return v_IP_SOCKET_IND.CTRL.ConnectionId.Local;

  }


Change 3

	Function name
	f_TcpClient_Connect

	Reason for change
	As defined in change 1, flag , tcpClientConnectInd needs to be passed. It is passed as ‘inout’ parameter as this value needs to be retained by calling function

	Summary of change
	Passed the flag tcpClientConnectInd to called function fl_TCP_Connect

	Source of change
	IP_PTC\TCP_Functions.ttcn


Before change:

	  function f_TcpClient_Connect(template (value) IP_Socket_Type p_LocalSocket,

                               template (value) IP_Socket_Type p_RemoteSocket,

                               IP_DrbInfo_Type p_DrbInfo,

                               InternetApplication_Type p_Application) runs on IP_PTC return template (value) IP_Connection_Type

  {

    var IP_Socket_Type v_LocalSocket;

    var template (value) IP_Connection_Type v_IP_Connection;

    v_LocalSocket := fl_TCP_Connect(p_LocalSocket, p_RemoteSocket, p_Application);
    v_IP_Connection := cs_TCP_ConnectionId(v_LocalSocket, p_RemoteSocket);

    // Modify routing table

    f_IP_PTC_RoutingTable_AddEntry(v_IP_Connection, p_DrbInfo);

    return v_IP_Connection;

  }


After change :

	  function f_TcpClient_Connect(template (value) IP_Socket_Type p_LocalSocket,

                               template (value) IP_Socket_Type p_RemoteSocket,

                               IP_DrbInfo_Type p_DrbInfo,

                               InternetApplication_Type p_Application,

                               inout boolean tcpClientConnectInd := false) runs on IP_PTC return template (value) IP_Connection_Type

  {

    var IP_Socket_Type v_LocalSocket;

    var template (value) IP_Connection_Type v_IP_Connection;

    v_LocalSocket := fl_TCP_Connect(p_LocalSocket, p_RemoteSocket, p_Application, tcpClientConnectInd);

    v_IP_Connection := cs_TCP_ConnectionId(v_LocalSocket, p_RemoteSocket);

    // Modify routing table

    f_IP_PTC_RoutingTable_AddEntry(v_IP_Connection, p_DrbInfo);

    return v_IP_Connection;

  }

